
Exposé: A Tree Pattern Function in 
Mathematica
For: “Wissenschaftliches Arbeiten,” Software Engineering at 
Fachhochschule Oberösterreich, Campus Hagenberg.

The present author has implemented the following  set of Wolfram Language functions in a recent 
Wolfram Community project (Heseltine, 2023) and would like to expand upon the underlying 
concepts and the language used, in an effort to further his own understanding and present a 
helpful guide to the language and its functional programming paradigm support.

In [19] := proofID[Grid[{___, {ID, id_}, ___}, ___]] := id;

subproofs[

Grid[{___, {Proofs, OpenerView[{Arguments, Column[subproofs_, ___]}, ___]}, ___},

___]] := subproofs;

subproofs[proof_] := {};

getLeanTree[proof_] := Tree[proofID[proof], getLeanTree /@ subproofs[proof],

TreeElementLabelStyle  All  Directive[White, 16, FontFamily  "Times New Roman"],

TreeElementStyle  All  Directive[EdgeForm[Black], RGBColor["#B6094A"]]]

An exposé of the concepts follows, drawing on all three functions (one function is overloaded to 
produce two forms of the same function), but first an introduction of both the problem space and 
language will be helpful. The complete code base is available as a G itHub repository.

The Problem Space
The objective is to extract a tree data structure in the form of certain integer mathematical proof 
IDs and the related children IDs from a grid expression in Wolfram Language. The grid is interpreted 
by Mathematica but comes from an external program called LEAN, an automated and interactive 
theorem prover. LEAN and the particular implementation details of processing the tree can be 
treated as a black box for the purposes of this study.

The Wolfram Language
Wolfram Research (2023) describes the language on a high level like this: “The Wolfram Language is 
a highly developed knowledge-based language that unifies a broad range of programming 
paradigms and uses its unique concept of symbolic programming to add a new level of flexibility to 
the very concept of programming.” Also, on the point of functional programming: “Functional 
programming is a highly developed and deeply integrated core feature of the Wolfram Language, 

made dramatically richer and more convenient through the symbolic nature of the language.”

The functional aspects of Wolfram Language to the point needed to understand the function of 
interest will be of particular import here, but less so the symbolic side. Mathematica is the concrete 
programming environment used to explore the concepts with examples.

Lists and Replacements
Going back to our object of interest, here is the first of the set of functions we are presently 
studying.

I n [ ] : = proofID[Grid[{___, {ID, id_}, ___}, ___]] := id;

They use lists (demarcated with curly brackets, here in a nested fashion) in constructing a pattern, 
which will be discussed in a later section, but lists can also be generated: Lists are central, general 
objects in the Wolfram Language since they can be made to represent other objects. 

Internally, they are, of course, functions.

I n [ ] : = FullForm[{1, 2, 3}]

Ou t [ ] / / F u l l F o rm=

List[1, 2, 3]

Starting with list creation utilities, here are some helpful functions for handling lists.

I n [ ] : = Range[3]

Ou t [ ] =

{1, 2, 3}

This is the same as:

I n [ ] : = Range[1, 3, 1]

Ou t [ ] =

{1, 2, 3}

Compare this to Table, which takes a Wolfram Language iterator in the form {i, imin, imax, step}:

I n [ ] : = Table[2 k, {k, 1, 10, 2}]

Ou t [ ] =

{2, 6, 10, 14, 18}

As with Range, imin and step can be omitted (set to 1).

One might like to change the display orientation, here using “%” to access the evaluation immedi-
ately prior:

I n [ ] : = Column[%]

Ou t [ ] =

2

6

10

14

18

Matrices are similarly a matter of display, of nested lists. A construction using two iterators might 
look as follows.

I n [ ] : = Table[i * j, {i, 1, 3}, {j, 1, 4}] // MatrixForm

Ou t [ ] / /Ma t r i x F o rm=

1 2 3 4

2 4 6 8

3 6 9 12

It is important to note the difference to this expression reversing the order of the iterators.

Table[i * j, {j, 1, 4}, {i, 1, 3}] // MatrixForm (* // TableForm *)

Ou t [ ] / /Ma t r i x F o rm=

1 2 3

2 4 6

3 6 9

4 8 12

That is, the first iterator determines the number of rows, in the 2D context. Another way to say this 
is that the inner iterator determines the value of the outer iterator. The mathematical expression 
being evaluated is, of course, commutative, however.

For measuring lists, the Length-Function gives the outermost dimension, whereas Dimensions[] 
gives all dimensions of nested lists. The number of dimensions is given by ArrayDepth[].

For testing and sampling lists the following functions are available. Position[list, element] gives the 
(list of) index positions at which element sits, whereas Select[list, predicate] tests elements of the 
list against predicate, sampling those elements that satisfy, say, even parity with EvenQ. Given a 
position, elements can be extracted using Part[] or its short form in double-square-brackets nota-
tion, here as a matrix-example.

I n [ ] : = Table[ai,j, {i, 3}, {j, 3}] // MatrixForm

Ou t [ ] / /Ma t r i x F o rm=

a1,1 a1,2 a1,3
a2,1 a2,2 a2,3
a3,1 a3,2 a3,3

I n [ ] : = %〚2, 1〛

Ou t [ ] =

a2,1

Take[] is similar to Part[] but allows sampling consecutively placed elements in a list.

This section ends with an analysis of the “_”-character to symbolize any other Mathematica object 
(Wolfram Language expressions), which is useful for patterns in general, and particularly, the 
proofID function at the beginning of this section. The crucial difference is between two under-
scores, “__” or BlankSequence, standing for any sequence of one or more expressions, and three 
underscores, “___” or BlankNullSequence, which also allows zero such expressions.

The following example, taken from Wolfram Research (2023), makes the limitation of BlankSe-
quence clear.

I n [ ] : = f[x__] := Length[{x}]

I n [ ] : = {f[x, y, z], f[]}

Ou t [ ] =

{3, f[]}

Contrast this with use of the BlankNullSequence:

I n [ ] : = f2[x___] := p[x, x]

I n [ ] : = {f2[], f2[1], f2[1, a]}

Ou t [ ] =

{p[], p[1, 1], p[1, a, 1, a]}

Blank (“_”) is the most limited pattern character in that it can stand for any Wolfram Language 
Expression, but only exactly one such expression.

I n [ ] : = f3[x_] := "evaluated"

I n [ ] : = {f3[], f3[1], f3[1, "a"]}

Ou t [ ] =

{f3[], evaluated, f3[1, a]}

Functional Programming
With lists there is now a background in place to consider Wolfram Language’s functional aspects in 
more detail. For a definition, I refer to the Mathematica Cookbook (Mangano, 2010, p. 31): “All 
functional languages emphasize the evaluation of expressions to produce values rather than 
commands or statements that are executed for their side effects.” Since Mathematica also allows 
for functions like Do[], the language is not purely functional, instead supporting at least also the 
procedural paradigm. The focus here however is functional programming by a given example, not a 
comparison of paradigms or more analysis of the functional one.

With this, the following core functional programming “primitives” are summed up in a helpful 
reference table in (Mangano, 2010, p. 26) and will lead to a discussion of the relative idioms and 
their application.

Out[107]=

Function Operator Description

Map[f, expr] /@ Return the list that results from

executing f on each element of an expr

Apply[f, expr] @@ Return the result of replacing

the head of a list with function f

Apply[f, expr, {1}] @@@ Applies f at level 1

inside list. In other words,

replace the head of all elements

Fold[f, x, {a1, a2, a3}] N/A If list hast length 0,

return x, otherwise return

f[f[f[x, a1], a2], a3]...

FoldList[f, x, {a1, a2, a3, ...}] N/A Return the list {x, f[x,

a1], f[f[x, a1], a2], ...}

Nest[f, expr, n] N/A Return the list f[f[f[...f[expr]...]]]

(i.e. f applied n times)

NestList[f, expr, n] N/A Return the list {x, f[expr], f[f[expr]],

...} where f repeats up to n times

This also covers almost all of the sort-hand notation after blanks, where ampersand (“&”) and hash 
(“#”) still need mentioning: this ties in perfectly with the topic of functional operations and goes to 

show that indeed, functional is at the heart of the language, since the short-hand notation offered 
caters to the paradigm. This example illustrates the usages, in so-called pure functions (lambda 
expressions), starting with the named-function counter-example.

In [14] := h[x_] := f[x] + g[x]

Combine this with Map[] in the following way.

In [15] := Map[h, {a, b, c}]

Out[15]=

{f[a] + g[a], f[b] + g[b], f[c] + g[c]}

The pure-function alternative is written as follows.

In [16] := Map[f[#] + g[#] &, {a, b, c}]

Out[16]=

{f[a] + g[a], f[b] + g[b], f[c] + g[c]}

Here ampersand demarcates the end of the function expression, and hash marks the arguments. 
The non-short-hand version is:

In [18] := Map[Function[x, f[x] + g[x]], {a, b, c}]

Out[18]=

{f[a] + g[a], f[b] + g[b], f[c] + g[c]}

I refer to the Functional Operators Tutorial (Wolfram Research, 2023) for the minor details includ-
ing, for example, the modifications of the hash operator, such as #n or ##. (Mangano, 2010) offer the 
helpful notion of idioms in this conext, where the functions Map[] and Apply[] that we have already 
seen present one such idiom, useful for summing sublists: here is the relevant example (p. 33), in 
short-hand.

In [23] := Plus @@ # & /@ {{1, 2, 3}, {4, 5, 6, 7, 8}, {9, 10, 11, 12}}

Out[23]=

{6, 30, 42}

Plus is applied and mapped to each element, here each sublist, hence this would be the Map-Apply 
idiom. The same can be accomplished in other ways.

In [24] := Plus @@@ {{1, 2, 3}, {4, 5, 6, 7, 8}, {9, 10, 11, 12}}

Out[24]=

{6, 30, 42}

(Mangano, 2010) goes on to present further use cases and appropriate idioms in the functional 
programming chapter of his book. The functional programming elements discussed so far allow us 
to analyze this part of the code, the third of the three functions, we are investigating: 

In [25] := getLeanTree[proof_] := Tree[proofID[proof], getLeanTree /@ subproofs[proof],

TreeElementLabelStyle  All  Directive[White, 16, FontFamily  "Times New Roman"],

TreeElementStyle  All  Directive[EdgeForm[Black], RGBColor["#B6094A"]]]

There are a lot of options here, concerned with output style. (Something like the following gets 
generated as the final return value: the presentation is controlled by the options.)

Ou t [ ] =

24
0

23
1

22
2

21
3

20
4

19

8
5

7
6 3

5

18 17
10

16
11

15

13
12 3

11
14

10

4

There is also a recursive element here, which we will ignore just for the moment, in fact we want to 
focus in on just this part: 

In [106] :=

getLeanTree /@ subproofs[proof]

The "/@" - Mapping of the (recursively called) function getLeanTree[] is the Map - Apply idiom in 
action, where the function is applied to the output of another function subproofs[], which is a list, 
with usually two or three elements, behind the scenes .

I n [ ] : = subproofs[leanProof]

Ou t [ ] =



Goal p : ℕ
ID 0

Rule Assumption

,

Goal ∀ {m n : ℕ}, nat.prime p → p ∣ m ^ n → p ∣ m

ID 23

Rule ∀I

Proofs Arguments

Goal m : ℕ
ID 1

Rule Assumption

Goal ∀ {n : ℕ}, nat.prime p → p ∣ m ^ n → p ∣ m

ID 22

Rule ∀I

Proofs Arguments



Rule-based Programming
The grid structures presented above immediately bring us to pattern-oriented programming in 
Wolfram Language and we move on to the second function of our initial three to investigate this 
aspect.

In [108] :=

subproofs[

Grid[{___, {Proofs, OpenerView[{Arguments, Column[subproofs_, ___]}, ___]}, ___},

___]] := subproofs;

subproofs[proof_] := {};

To clear up a technical detail: the opener view is the collapsable arrow or triangle shape leading to 
more text display, here in a nested form when viewing in a Mathematica notebook. In a way, the 
pattern to tree functions are all about this un-collapsing of the grid structure, representing a mathe-

matical proof in Lean. For completeness, here is another level of the grid expanded and viewable in 
PDF-file-format to illustrate the functionality. (This structure will become important again once we 
look at recursion.)

Ou t [ ] =



Goal p : ℕ
ID 0

Rule Assumption

,

Goal ∀ {m n : ℕ}, nat.prime p → p ∣ m ^ n → p ∣ m

ID 23

Rule ∀I

Proofs Arguments

Goal m : ℕ
ID 1

Rule Assumption

Goal ∀ {n : ℕ}, nat.prime p → p ∣ m ^ n → p ∣ m

ID 22

Rule ∀I

Proofs Arguments

Goal n : ℕ
ID 2

Rule Assumption

Goal nat.prime p → p ∣ m ^ n → p ∣ m

ID 21

Rule ∀I

Proofs Arguments



With this out of the way, rules are the essential building block in this chapter: “Everything that the 
Wolfram Language does can be thought of as derived from its ability to apply general transforma-
tion rules to arbitrary symbolic expressions” (Wolfram Research, 2023). They can be used for string 
operations like so, where the arrow gives the replacement rule.

In [115] :=

StringReplace["Mathematica is multi-paradigm", {"Mathematica"  "Wolfram Language"}]

Out[115]=

Wolfram Language is multi-paradigm

Here is short-hand notation and a useful list item swap rule:
In [117] :=

{{α, 1}, {β, 2}, {γ, 3}} /. {x_, y_}  {y, x}

Out[117]=

{{1, α}, {2, β}, {3, γ}}

At this point we also circle back to the blank-notation for patterns: The above example is taken 
from (Wellin, p. 150), and to highlight the concept’s importance and its relationship to lists: “Rule-
based programming is such a useful construct for manipulating lists and arbitrary expressions that 
no user of Mathematica should be without a working knowledge of this paradigm.” (Ibid.)

The following built-in functions help to implement the paradigm as discussed so far: Here a generic 
test using blanks.

In [118] :=

MatchQ["Mathematica", _]

Out[118]=

True

And here the pattern matching function already discussed in terms of lists, the first of our three 
functions, implementing a more complex pattern matching routine.

I n [ ] : = proofID[Grid[{___, {ID, id_}, ___}, ___]] := id;

Combine this pattern with the built-on Cases[] to extract the matches and compile them to a list, 
given appropriate input.

In [120] :=

pattern = Grid[{___, {ID, id_}, ___}, ___];

In [127] :=

Cases[subproofs[leanProof], pattern]

Out[127]=



Goal p : ℕ
ID 0

Rule Assumption

,

Goal ∀ {m n : ℕ}, nat.prime p → p ∣ m ^ n → p ∣ m

ID 23

Rule ∀I

Proofs Arguments



In this case, Cases[] simply returns the input list, because all elements match. A more useful applica-
tion of cases might be:

In [129] :=

Cases[{"here is a diverse list", -2, 2.0, 0, -0, -0.0, -0.1}, _?Negative]

Out[129]=

{-2, -0.1}

Here the negative-test (a predicate function, instead of a pattern) is applied, where Negative[] 
returns true in the appropriate case and false (or unevaluated, if the predicate is not applicable) in 
all others.

In [132] :=

Negative /@ {-2, -0.1}

Out[132]=

{True, True}

In [134] :=

Negative /@ {"here is a diverse list", 2.0, 0, -0, -0.0}

Out[134]=

{Negative[here is a diverse list], False, False, False, False}

On the topic of predicates, pure functions can also be used to implement them, here in combina-
tion with a test for the head of an element (testing if it is a list).

In [139] :=

MatchQ[{1, 2, 3}, _List?(Length[#] > 2 &)]

Out[139]=

True

Finally, the second of our functions does the same thing as the first, but for an even more complex 
pattern (again using BlankSequences liberally).

In [140] :=

subproofs[

Grid[{___, {Proofs, OpenerView[{Arguments, Column[subproofs_, ___]}, ___]}, ___},

___]] := subproofs;

subproofs[proof_] := {};

But with one caveat of a base case for a later recursion, in the form of an overloaded function 
(subproofs[]) taking any possible pattern and binding it to the variable proof, but returning an 
empty list as a fixed value: this leads to the final topic to be discussed in this analysis, recursion.

Recursion
We now can put the functions together to explain the functionality in terms of recursion, another 
paradigm implemented in Wolfram Language. This is the set of functions we were considering:

proofID[Grid[{___, {ID, id_}, ___}, ___]] := id;

(* basically an extractor-function, using patterns *)

subproofs[Grid[{___,

{Proofs, OpenerView[{Arguments, Column[subproofs_, ___]}, ___]}, ___}, ___]] :=

subproofs; (* Same as above, but with more complex patterns *)

subproofs[proof_] := {}; (* Stopping condition preparing the recursion *)

getLeanTree[proof_] := Tree[proofID[proof], getLeanTree /@ subproofs[proof],

TreeElementLabelStyle  All  Directive[White, 16, FontFamily  "Times New Roman"],

TreeElementStyle  All  Directive[EdgeForm[Black], RGBColor["#B6094A"]]]

(* Putting it all together now, plus some options for styling the output *)

One note on Proofs and Arguments: these are simply the names of heads in the proof-grid-structure 
that is the input for the main function, getLeanTree[].

Recursion is of course a call of a function to itself, which is seen here in getLeanTree[], as the func-
tion constructs a tree with an integer delivered by proofID as the parent and the children as outputs 
from the recursive call mapped to the proofs returned by subproofs[]. Say two children are 
returned: the next recursion step makes a tree out of the “child proofs” (the “sub-proofs”) and 
again extracts the ids and makes the recursive call, now implementing the branching of the tree at 
the second level. 

The fact that subproofs[] is is overloaded to produce an empty list when no relevant pattern is 
present ensures that in the recursive mapping of getLeanTree[] eventually an empty list is mapped 
to:

In [143] :=

Tree["id", {}]

Out[143]=

id

A leaf-node is reached, here corresponding to a mathematical axiom which in turn does not depend 
on other axioms: The recursion stops, a base case is reached in the sense that a tree with empty 
children is defined as a leaf, though the actual empty list is returned not by the recursing function 
(getLeanTree[]) but by another (subproofs[]).

This section closes with a more canonical example of recursion with a direct base case, the 
Fibonacci sequence, and an application to lists, showing how lists are once again the essential 
auxiliary concept.

In [157] :=

f[1] = 1; (* Fibonacci *)

f[2] = 1;

f[n_] := f[n - 2] + f[n - 1] /; n > 2

It turns out specific rules are looked up before more general rules, which is helpful for implement-
ing recursion, since we can omit the non-base condition after simply defining the base cases.

In [163] :=

f[n_] := f[n - 2] + f[n - 1]

In [164] :=

Table[f[i], {i, 15}]

Out[164]=

{1, 1, 2, 3, 5, 8, 13, 21, 34, 55, 89, 144, 233, 377, 610}

The following helpful function lets us analyze what is being evaluated in what order; this can be 
especially helpful when it comes to recursion, often implementing many steps in relatively sparse 
syntax.

In [169] :=

TracePrint[f[3]]

f[3]

f

3

RuleCondition[$ConditionHold[$ConditionHold[f[3 - 2] + f[3 - 1]]], 3 > 2]

RuleCondition

3 > 2

Greater

3

2

True

RuleCondition[$ConditionHold[$ConditionHold[f[3 - 2] + f[3 - 1]]], True]

$ConditionHold[$ConditionHold[f[3 - 2] + f[3 - 1]]]

$ConditionHold

f[3 - 2] + f[3 - 1]

Plus

f[3 - 2]

f

3 - 2

Plus

3

-2

1

f[1]

1

f[3 - 1]

f

3 - 1

Plus

3

-1

2

f[2]

1

1 + 1

2

Out[169]=

2

The final (simple!) list-recursion example echoes getLeanTree[] in the way it requires an empty list 
base case and recursively folds into itself.

In [170] :=

l[list_] := l[Rest[list]] + 1 (* length of a list is rest (tail) + 1 *)

l[{}] := 0

In [172] :=

l[{1, 2, 3, 4, 5}]

Out[172]=

5

The observation here is that a simple problem can actually appear more complex, simply because 
various paradigms and potentially some complex patterns are being mixed at the same time.

Limitations and Summary
This study did not cover numerics, graphics programming,  front-end programming, or writing 
packages, all major topics related to the current discussion: An Introduction to Programming with 
Mathematica (2005, Wellin et al.) completes the picture at the intended overview level of analysis, 
of exactly these missing topics. The focus here has been on one particular example, solving one 
problem, in Wolfram Language; it is involved enough to be viewed in context of the major themes 
chosen after introducing the language in broad strokes. These themes were lists; functional pro-
gramming; rule-based programming as the particularly helpful approach in this situation; and 
recursion, which we could not have easily done without.

